Pensieve header: Drawing the Willerton Fish on April 6, 2016.

The Willerton Fish

Willerton: It is amuzing to plot the values of v_2 against the values of v_3 on a large sample of knots.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
GD[K_] :=
  GD @@ PD[K] /. X[i_, j_, k_, l_] ⧴ If[PositiveQ[X[i, j, k, l]], Ap[l, i], Am[j, i]];
Column[GD /@ AllKnots[{3, 7}]]
```

KnotTheory::loading : Loading precomputed data in PD4Knots`.

```
GD[Am[4, 1], Am[6, 3], Am[2, 5]]
GD[Ap[1, 4], Ap[5, 8], Am[3, 6], Am[7, 2]]
GD[Am[6, 1], Am[8, 3], Am[10, 5], Am[2, 7], Am[4, 9]]
GD[Am[4, 1], Am[8, 3], Am[10, 5], Am[6, 9], Am[2, 7]]
GD[Am[4, 1], Am[10, 7], Ap[8, 3], Ap[2, 9], Am[12, 5], Am[6, 11]]
GD[Am[4, 1], Am[10, 5], Ap[8, 3], Ap[2, 9], Am[12, 7], Am[6, 11]]
GD[Ap[1, 4], Ap[3, 8], Am[9, 12], Am[5, 10], Am[11, 6], Ap[7, 2]]
GD[Am[8, 1], Am[10, 3], Am[12, 5], Am[14, 7], Am[2, 9], Am[4, 11], Am[6, 13]]
GD[Am[4, 1], Am[10, 3], Am[14, 5], Am[12, 7], Am[8, 11], Am[6, 13], Am[2, 9]]
GD[Ap[1, 6], Ap[3, 10], Ap[7, 14], Ap[13, 8], Ap[5, 12], Ap[9, 2], Ap[11, 4]]
GD[Ap[1, 6], Ap[5, 12], Ap[7, 14], Ap[13, 8], Ap[11, 2], Ap[3, 10], Ap[9, 4]]
GD[Am[4, 1], Am[10, 3], Am[12, 5], Am[14, 7], Am[6, 13], Am[8, 11], Am[2, 9]]
GD[Am[4, 1], Am[8, 3], Am[12, 5], Ap[14, 9], Ap[10, 13], Am[6, 11], Am[2, 7]]
GD[Am[4, 1], Am[10, 5], Ap[8, 3], Ap[2, 9], Am[14, 11], Ap[12, 7], Ap[6, 13]]
```

```
Draw[gd_GD] := Module[{n = Length@gd}, Graphics[{
    Line[{{0, 0}, {2 n + 1, 0}}],
    List @@ gd /. (ah_)[i_, j_] ⧴ {
       Arrow[BezierCurve[{{i, 0}, 0.5 {i + j, Abs[j - i]}, {j, 0}}]],
       Text[ah /. {Ap → "+", Am → "-"}, {j, 0.3}]},
    Table[Text[i, {i, -0.5}], {i, 2 n}]
   }]]
```

```
Draw[GD[Ap[3, 1], Am[2, 4]]]
```



```
Draw /@ GD /@ AllKnots[{3, 7}]
```



From Polyak-Viro's "Gauss Diagram Formulas for Vassiliev Invariants", IMRN 11 (1994) 445-453:

3.B THEOREM 1. *If G is any based Gauss diagram of a knot K then*

(4) $$v_2(K) = \left\langle \quad , G \right\rangle,$$

where $v_2(K)$ is the Vassiliev invariant of degree 2 which takes values 0 on the unknot and 1 on a trefoil.

```
Subsets[{a, b, c, d}, {2}]
```

```
{{a, b}, {a, c}, {a, d}, {b, c}, {b, d}, {c, d}}
```

```
GD /: Normalize[gd_GD] := SortBy[
    gd /. Thread[Union @@ (List @@@ gd) → Range[2 Length[gd]]],
    (List @@ #) &
  ];
GPV[F_GD, K_GD] /; Length[F] > Length[K] := 0;
GPV[F_GD, K_GD] /; Length[F] < Length[K] := Total[
    GPV[F, #] & /@ Subsets[K, {Length[F]}]
  ];
GPV[F_GD, K_GD] /; Length[F] == Length[K] := If[
    F =!= Normalize[K /. Ap | Am → A], 0, (-1) ^Count[K, _Am]
  ];
V2[K_] := V2[K] = GPV[GD[A[2, 4], A[3, 1]], GD[K]];
```

```
V2 /@ AllKnots[{3, 8}]
```

```
{1, -1, 3, 2, -2, -1, 1, 6, 3, 5, 4, 4, 1, -1, -3, 0,
 -4, -3, -1, -2, 2, 2, -2, 3, -1, -3, 1, 0, 4, 1, -1, 1, 5, 2, 0}
```

From Polyak-Viro’s “Gauss Diagram Formulas for Vassiliev Invariants”, IMRN 11 (1994) 445-453:

4.A Theorem 2. *If G is a Gauss diagram of a knot K then*

(5) $$v_3(K) = \left\langle \frac{1}{2}\left[\,\cdot\,\right] + \left[\,\cdot\,\right], G \right\rangle,$$



where $v_3(K)$ is the Vassiliev invariant of degree 3 which takes values 0 on the unknot, +1 on the right trefoil and −1 on the left trefoil.

```
GPV[F_Plus, K_] := GPV[#, K] & /@ F;
GPV[c_ * F_GD, K_] := c GPV[F, K];
GPV[0, K_] = 0;
GD /: RotateLeft[gd_GD, k_Integer] := Normalize[
    gd /. i_Integer ⧴ Mod[i - k, 2 Length@gd, 1]
  ];
V3Ds = Expand[Plus[
     1/2 Sum[RotateLeft[GD[A[1, 5], A[4, 2], A[6, 3]], k], {k, 0, 5}],
     Sum[RotateLeft[GD[A[1, 4], A[5, 2], A[3, 6]], k], {k, 0, 1}]
    ]];
V3[K_] := V3[K] = GPV[V3Ds, GD@K];
```

```
V3 /@ AllKnots[{3, 8}]
```

```
{-1, 0, -5, -3, 1, 1, 0, -14, -6, 11, 8, -8, -2, -1,
 3, 1, 0, 1, -3, 3, 2, 1, 0, 3, 2, 0, 1, 0, -7, -1, 0, 0, 10, -2, 1}
```

```
Ks = AllKnots[{3, 9}]; Ks = Ks ⋃ (Mirror /@ Ks);
ListPlot[{V2[#], V3[#]} & /@ Ks, PlotStyle → {Red}, Axes → False, PlotRange → All]
```



```
Ks = AllKnots[{3, 10}]; Ks = Ks ⋃ (Mirror /@ Ks);
Histogram3D[{V2[#], V3[#]} & /@ Ks, {1}, ChartElements → Graphics3D[Cylinder[]]]
```